

# Z80-PIO的非中断使用

朱庆华

(自动化系)

**摘要** 本文介绍了 Z80-PIO 的非中断使用, 列举一些输入输出的特殊用法, 使 Z80-PIO 的使用更为完善。

**主题词** 非中断使用; 输入; 输出

## 0 概述

Z80-PIO 是 Z80 系统中最重要的接口, 它具有 2 个独立的八位二进制平行输入/输出, 即口 A 与口 B, 如图 1 所示。口 A 或口 B 可作输入、输出、双向(仅口 A 有)和位控 4 个用途。每个口又分为控制口与数据口, Z80-PIO 是一个软硬结合的可编程芯片, 可编程的意思是使用该芯片的口 A 或口 B 之前要对其控制口写入控制字以规定该口的用途, 然后才能使用相应的数据口, 以实现规定其用途。数据口每个口均有 8 根输入/输出线, 它们与 TTL 兼容。以最常用的单板机 TP801A 中的 Z80-PIO 为例, 其口 A、口 B 的控制口地址分别为 82H、83H, 数据口地址分别为 80H、81H。

目前现有的微机类书籍, 讲到 Z80-PIO 时, 几乎未提及在非中断情况下的使用, 因为“Z80-PIO 有一个突出的特性”, 外部设备与 CPU 之间的全部数据传送是在中断控制之下实现的。因而使用 Z80-PIO 时常要写入 3 或 5 个控制字。笔者认为: Z80-PIO 在中断情况下使用是由于中断服务程序占用 CPU 时间外, 其他时间 CPU 可干另外的事。但在简单控制输入/输出场合, Z80-PIO 非中断使用也是完全可能的。可省去 2 个控制字, 即中断矢量字与中断允许字。仅需写 1 个或 3 个控制字到控制口, 又能使相应数据口实现同样的功能。因而 Z80-PIO 的使用, 即可在中断情况下进行, 又可在非中断的情况下进行。以下是 Z80-PIO 的非




图 1 引脚图

本文1990年6月22日收到

中断输入输出用法。

## 1 非中断输出

假使要求口B的PB0呈高电平, 即PB0 = 1, 其程序如下:

```

                ORG          2000H
3E0F  LD      A, 0FH      ; 输出控制字→A
D383  OUT    (83H), A    ; A→口B控制口
3E01  LD      A, 01H    ; 01H→A
D381  OUT    (81H), A    ; A→口B数据口
76    HALT   ; 暂停
  
```

PB0 = 1 意味着经晶体三极管驱动可点亮一盏灯如图 2 A 所示, 或经光电隔离使某继电器线圈 J 通电如图 2 B 所示, 从而使 J 的触点闭合或断开, 去控制电动机或其他用电设备的启动或停止。改变上述 01 的内容, 可以在不同的位进行类似的控制。同理, 口 A 的使用基本上同口 B, 仅将上述程序的口地址改变一下, 即可实现。




图2 非中断所输出

使口B的PB0—PB78根线间隔 1 min 轮换出现高电平, 程序如下:

```

                ORG          2010H
3E0F  LD      A, 0FH
D383  OUT    (83H), A
3E01  LD      A, 01H
D381  LOOP:  OUT(81H), A
CD2020 CALL   DELAY      ; 调用延时 1 min 子程序
07    RLCA   ; A 循环左移一位
18FB  JR     LOOP      ; 转 LOOP

                ORG          2020H
063C  DELAY: LD   B, 3CH ; B←60
C5    LOOP2: PUSH BC ; BC 进线保护
0632  LD     B, 32H ; B←50
  
```

```

CD4F06LOOP1:CALL    D20ms    ; 调用20ms子程序
10FB          DJNZ    LOOP1    ; B-1≠0转LOOP1
C1           POP     BC        ; 出栈,恢复BC原值
10F5          DJNZ    LOOP2    ; B-1≠0转LOOP2
C9           RET              ; 返回
    
```

子程序延时时间为1 min, 因为内循环 LOOP1 是 50 次调用 20ms, 花去 1s, 外循环 LOOP2 是60次做内循环, 因而完成DELAY的时间约 1 min。

本主程序口B 8 根可作顺序控制器的 8 道工序驱动源, 如图 3 所示, n 根垂直线连接被驱动电路, 口B的 8 根线, 按工艺控制要求, 经二极管连接垂直线(即为矩阵译码电路)就可实现顺序控制。如图中第 1 道工序 J<sub>n</sub>动作, 第 2 道工序 J<sub>1</sub>、J<sub>n</sub>动作, 第 3 道工序 J<sub>2</sub> 动作, 第 4 道工序 J<sub>1</sub>、J<sub>3</sub>动作……, 延时时间的快慢由内循环次数与外循环次数决定。




图 3 顺序控制驱动电路图

这里工序仅 8 道, 若口A、口B同时使用, 则工序可以多达 16 道。假使控制对象不多, 如 8 个被控对象, 但工序很多如有 50 道工序, 这时口B 8 根线, 可以直接连驱动电路, 而每道工序的要求分别放入内存如 2500H—2531H的话, 程序可以改为如下

```

          ORG          2030H
3E0F     LD           A, 0FH
D383     OUT          (83H), A
210025   LOOP4: LD     HL, 2500H    ; HL 指向工序要求首址
0632     LD           B, 32H      ; 50道工序送B
C5       LOOP3: PUSH  BC         ; 保护BC
E5       PUSH       HL         ; 保护HL
7E       LD           A, (HL)    ; A←工序要求
D381     OUT          (81H), A    ; A→数据口B
    
```

CD2020	CALL	DELAY	; 调延时子程序
E1	POP	HL	; 恢复HL
23	INC	HL	; HL ← HL + 1
C1	POP	BC	; 恢复BC
10F3	DJNZ	LOOP3	; B - 1 ≠ 0 转 LOOP3
18EC	JR	LOOP4	; 转 LOOP4

本程序工序控制可多达256道, 只要将指令LD B, 32H改为LD B, 00H即可, 因为DJNZ LOOP3是B先减1后判断B是否为0, 而十六进制数00H - 01H = FFH故为256道工序。当然若要多于256道工序的话, 可用指令LD BC, nn代LD B, 32H, nn为工字节十六进制数, 这时DJNZ LOOP3要改为DEC BC, LD A, C, OR A, B, JR NZ, LOOP3 4条指令。

## 2 非中断输入

从口A取数, 并存2500H单元程序如下:

	ORG	2500H	
3E4F	LD	A, 4H	; 输入控制字送A
D382	OUT	(82H), A	; A → 控制口A
DB80H	IN	A, (80H)	; 从口A取数
320025	LD	(2500H), A	; 存数在2500H
76	HALT		

与输出一样仅写入1个控制字, 即可使用其数据口, 采进来的数存入2500H单元, 2500H单元内容或八位数代表了口A数据线PA7—PA0的高、低电平。同理口B的使用基本同口A, 仅将上述程序的口地址改变为口B的口地址即可。

用口A的PA0线对外界发生的脉冲进行计数, 结果存入2500H—2502H, 并显示其结果。连接如图4所示, 用了RS触发器组成消除按钮抖动的电子电路作为模拟脉冲发生。

外界发生脉冲引到PA0线上要满足外界脉冲宽度大于采样间隔时间, 否则会漏计, 同时按外界脉冲唯一的上升沿或下降沿进行计数, 否则会多计。一般情况下外界脉冲宽度大于采样间隔时间, 用上升沿或下降沿计数是一样的, 我们采用上升沿计数, 将上次采样为0, 本次采样为1, 表示有了上升沿可以计数。我们让上次信息放在E中, 本次信息在A中, 判断时将A与E分别右移一位, 使其分别进入CY位, 再判断是否1或0, 来决定是否上升沿; 程序流程如图5所示, 程序如下:




图4 口A输入PI0

```

ORG 2060H
210025 LD HL, 2500H ; 计数单元2500H
3600 LD (HL), 00H ; 2502H清0
23 1NC HL
3600 LD (HL), 00H
23 1NC HL
3600 LD (HL), 00H
1E00 LD E, 00H ; 上次信息存放E清0
3E4F LDA, 4FH ; 输入控制字送控制口A
D382 OUT (82H), A
DB80 LOOP5: 1N A, (80H); 口A取数
57 LD D, A ; 存D
OF RRCA ; A循环右移一位
301D JR NC, DONE ; 本次信息为0, 转
CB3B SRL E ; 本次为1则E右移一位
3819 JR C, DONE ; 上次为1转
210025 LD HL, 2500H ; 上升沿计数
34 1NC (HL) ; 并调整成十进制数
7E LD A, (HL)
27 DAA
77 LD (HL), A
FE00 CP 00H ; 与0比较
200E JR NZ, DONE ; 非0转
23 1NC HL ; 为0, 则进位
34 1NC (HL) ; 加1, 并调整成十进制数
7E LD A, (HL)
27 DAA
77 LD (HL), A
FE00 CP 00H
2005 JR NZ, DONE
23 1NC HL
34 1NC (HL)
7E LD A, (HL)
27 DAA
77 LD (HL), A
5A DONE: LD E, D ; 保存本次信息
CDB020 CALL DISUP ; 调用显示程序
18D7 JR LOOP5

```




图5 程序计数流程图

```

                                ORG    20B0H
D9      DISUP : EXX                ; 保护现场
210025          LD    HL, 2500H    ; 计数单元内容送显示缓冲器
DD21FB2F       LD    IX, 2FFBH
0603          LD    B, 03H
7E      LOOP6 : LD    A, (HL)
CD0021          CALL  UFI
10FA          DJNZ  LOOP6
21F72F          LD    HL, 2FF7H    ; HL指向显示缓冲器首址
0620          LD    B, 20H        ; B指向左端灯位
5E      DISI : LD    E, (HL)
1600          LD    D, 00
3E00          LD    A, 00
D38C          OUT   (8CH), A      ; 送显示
DD21A607       LD    IX, 07A6H    ; IX指向字形表首址
DD19          ADD   IX, DE        ; 求得字形代码地址
DD7E00         LD    A, (IX + 00)  ; 获得字形代码
D388          OUT   (88H), A      ; 送字形口
78           LD    A, B          ; 灯位代码送灯位口
D38C          OUT   (8CH), A
1E2D          LD    E, 2DH        ; 延时 1 ms
1D      DISZ : DEC   E
3E00          LD    A, 00
BB           CP    E
20FA          JR    NZ, DISZ      ; 显示延时1ms
3E01          LD    A, 01H
B8           CP    B            ; B = 1码
2805          JR    Z, DIS3       ; 是转
23           INC   HL            ; 否, 指向下一位
CB38          SRL   B            ; 灯位右移一位
18D9          JR    DIS1         ; 转显示下一位
AF      DIS3 : XOR   A          ; 关显示
D38C          OUT   (8CH), A
D9           EXX
C9           RET

                                ORG    21C0H
CD3C06  UF1 : CALL  UF0R1        ; 调用将A高4位→(IX + 00);
                                           低4位→(IX + 01)子程序

```

```

DD2B    DEC    IX
DD2B    DEC    IX
23      INC    HL
C9      RET
    
```

本程序执行后，一旦有外界脉冲输入，就有计数并显示功能，输入脉冲由光、电、机械等转换而成均可，程序计数为3个单元，采用十进制数显示，故要进行十进制数调整，每次采样后由于调用显示程序需要花去6ms，这约为采样间隔时间，若不用显示，则也需要延时，否则采样间隔太快，甚至小于外界脉冲的上升沿，结果反而无法捕捉上升沿。因此一般需要采样间隔时间大于上升沿时间才行。

下面例子可同时计8个外界脉冲源的数，并将结果存入2500H—2507H单元，程序流程如图6所示。

程序如下：

```

ORG      2110H
210025  LD    HL, 2500H ; 计数单元2500—
                2507H清0
0608    LD    B, 08H ; 清0
AF      XOR  A
77 L1:   LD    (HL), A
23      INC  HL
10FC    DJNZ L1
5F      LD   E, A ; E清0
3E4F    LD   A, 4FH ; 口A作输入
D382    OUT (82H), A
DB80    LOOP8:INA,(80H); 取口A的数
210025  LD   HL, 2500H
0608    LD   B, 08H ; 循环8次
0F      LOOP7: RRCA ; A循环右移一位
300E    JR   NC, DONE1 ; CY=0转
CB3B    SRL  E ; CY=1, E右移一位
3801    JR   C, DONE2 ; CY=1转
34      INC (HL) ; 上升沿计数
23      DONE2: INC HL ; 地址+1
10F5    DJNZ LOOP7 ; B-1≠0转
5F      LD   E, A ; 保存本次信息
D9      EXX
CD4F06  CALL D20ms ; 延时
D9      EXX
18E6    JR   LOOP8
    
```




图6 口A同时计8路脉冲流程图

CB38 DONE1: SRL E  
18FI JR DONE2

本程序基本上类似于 PA0 计数程序, 由于 PA7—PA0 均计数, 故要解决每采样一次需要循环 8 次判断是否上升沿, 而且要确保本次信息 A 与上次信息 E 的位要一一对应。为此, 本次信息右移一位判定为 0 时, 尽管不用再判上次信息的位, 但上次信息存放的 E 则要右移一位。目的确保位的对应。当然右移或左移都是一样的。由于本次信息在 A 中, A 内容循环右移 8 次又恢复原内容, 故采样后 A 内容不作保护。为了方便, 本程序采样间隔调用 20ms 延时子程序。

### 参 考 文 献

- 1 周明德. 微型计算机硬件软件及其应用. 清华大学出版社, 1982
- 2 黄学祥, 王嘉文. 微处理机控制基础. 交大出版社, 1989
- 3 张秀琼, 吴定荣. 微型计算机原理及应用基础. 北京科技出版社, 1984
- 4 王义方. 微型计算机原理及应用. 机械工业出版社, 1984
- 5 Lance A. Leventhal. Z80 汇编语言设计. 科技文献出版社, 重庆分社 1981

## Z80-PIO Without Using Interrupt

Zhu Qinghua

(Dept. of Auto.)

**Abstract** A new method of operating Z80-PIO without using interrupt is introduced. Sample programs of I/O, employing both methods, with and without interrupt, for more satisfactory Z80-PIO operation are presented.

**Keywords** without using interrupt, input, output